September 11, 2019

Problem solved: make a consistent connection of long time normal diffusion as e.g. measured by NMR to some sublinear diffusion regime.

Conditions

1. Gaussian approximation is valid
2. sub/super-linear diffusion at short time is described by $\langle r^2(t)\rangle \propto t^\beta$
3. at some transition time t^* (tstar) the short time diffusion smoothly meets the long time diffusion
4. the condition that value and slope of $\langle r^2(t)\rangle$ match at $t*$ implies an offset for the mean square displacement in the long time behaviour.
5. besides the streching exponent β, tstar (t^*) is the only additional free parameter.

After a few operation we the arrive at:

$$\langle r^2(t < t^*)\rangle = 6\,\frac{t^\beta t^{*-\beta+1} D_{nmr}}{\beta} = 6\,\frac{(t/t^*)^\beta\ (t^* D_{nmr})}{\beta} \tag{1}$$

and

$$\langle r^2(t > t^*)\rangle = 6\,\frac{D_{nmr}\ (\beta\, t - \beta\, t^* + t^*)}{\beta} = 6\, D_{nmr}[t + t^*(1/\beta - 1)] \tag{2}$$

the intermediate scattering (factor) then follows immediately from the Gaussian approximation.

$$S(Q,t) = \exp(-q^2\,\langle r^2(t)\rangle/6) \tag{3}$$

M. Monkenbusch, JCNS-1 , July 2015

Subroutine:

```
function diffusion_beta(t,q,Dlong,tstar,beta) result(sqt)
!-------------------------------------------------------------------------------
!
! diffusion scattering function under Gaussian approximation
! but with subdiffusive start
!
! the physical parameters are the classic long time diffusion constant Dlong
!                          and a transition time tstar
!                          below tstar the diffusion is sub (super?) diffusive
!                          with <r**2>(t) ~ t**beta
!
! the transition is smooth (stetig 1 x differenzierbar) at tstar
! this implies that the long time diffusion is linear with an offset, which
! implicitly is controlle by the value of tstar and the smoothness condition
!
! Michael Monkenbusch, JCNS-1, June 2015
!-------------------------------------------------------------------------------
implicit none


double precision, intent(in)  :: t       ! time in the time units used in the diffusion constant units
double precision, intent(in)  :: q       ! "q", momentum transfer in the length units used for Dlong
double precision, intent(in)  :: Dlong   ! long time diffusion constant in units consistent wit t and q
double precision, intent(in)  :: tstar   ! transition time between short and long time diffusion (units as t)
double precision, intent(in)  :: beta    ! streching exponent for the sublinear initial part of diffusion

double precision              :: sqt     ! return value = intermediate scattering factor of the diffusion

double precision              :: r2


if(t < tstar) then
   r2  = 6*t**beta*tstar**(-beta+1)*Dlong/beta
else
   r2  = 6*Dlong*(beta*t-beta*tstar+tstar)/beta
endif

sqt = exp(-q*q*r2/6d0)

! testing
! write(6,'(3F18.9)') t, r2, sqt
! testing

end function diffusion_beta
```